```
'   PET hop sequence generator

'   Pinnacle technologies
'   J.W. MacConnell
'   03-03-2000
'   Version 1.2
    ver$ = "1.2"

    DO
        CLS
        indent% = 5
        LOCATE 3, 1
        PRINT TAB(indent%); "The output data file name is:  hoptable.txt "
        PRINT
        PRINT TAB(indent%); "Enter a seed value between -32,767 and 32,767."
        PRINT TAB(indent%); "A given seed value will always yield the"
        PRINT TAB(indent%); "same random sequence. ";

        INPUT i$

        IF ABS(VAL(i$)) <= 32767 THEN
            EXIT DO
        ELSE
            SOUND 700, 2
        END IF
    LOOP

    'Select the sequence
    seedval% = VAL(i$)
    RANDOMIZE (seedval%)

    DIM table%(255, 24), UsedFreq%(24)
    FOR seq% = 0 TO 255
        PRINT "Working on sequence # "; seq%

        'reinitialize used table
        FOR hop% = 0 TO 24
            UsedFreq%(hop%) = 0
        NEXT hop%

        FOR hop% = 0 TO 24
            DO
                'Pick a random hop
                potentialhop% = RND(1) * 24

                'See if the proposed hop is already used
                IF UsedFreq%(potentialhop%) <> 1 THEN
                    'It is not used
                    UsedFreq%(potentialhop%) = 1
                    table%(seq%, hop%) = potentialhop%
                    EXIT DO
                END IF
            LOOP
        NEXT hop%
    NEXT seq%

    'Now, print out the results to a file
```

```
OPEN "hoptable.txt" FOR OUTPUT AS #1

PRINT #1, "Seed Value Selected = "; LTRIM$(RTRIM$(STR$(seedval%)))
PRINT #1, "Program Version = "; ver$
PRINT #1,

FOR seq% = 0 TO 255
    SELECT CASE seq%
        CASE IS >= 100
        i$ = ""

        CASE IS >= 10
        i$ = " "

        CASE ELSE
        i$ = "  "
    END SELECT

    PRINT #1, "Sequence "; i$; LTRIM$(RTRIM$(STR$(seq%))); ": ";
    FOR hop% = 0 TO 24
        PRINT #1, LTRIM$(RTRIM$(STR$(table%(seq%, hop%))));
        IF hop% <> 24 THEN
            PRINT #1, ",";
        ELSE
            PRINT #1,
        END IF
    NEXT hop%
NEXT seq%

CLOSE #1
END
```